

# Equivalence Scales

## Goal

In order to get measures of poverty and/or income inequality in a population, it is necessary to compare income across different types of households. It is not logical to directly compare total household income between households of different sizes and composition.

Suppose you observe three levels of income (A, B, and C), where  $A > B > C$ . You cannot state that a household earning A is better off than one earning B unless you know the two households are similar in composition. For example, a family of 4 adult members receiving A is not necessarily better off than a couple with 2 children who receive B, and the family receiving B may not be better off than the childless couple receiving C.

For this reason, total household income needs to be adjusted to make it comparable across different households. This exercise gives one example of “equalizing” households using one specific equivalence scale.

## Activity

Summarise total disposable income, per capita disposable income, and equivalised disposable income using the “LIS equivalence scale” (i.e., the square root of the number of household members) in Finland in 2000. First calculate the averages for the total population. Then recalculate the same averages by the number of household members. Print your results only for households with 7 or fewer household members. Be sure to eliminate observations with zero or missing *dpi* and to use the appropriate weights.

## Guidelines

- Do not forget to “clean” the data. As always, it is important to be vigilant about missing values. Prior to Wave V, no distinction was made between 0 and missing values. Starting from Wave V, the lissification process consistently coded missing values with a “dot” and genuine 0 values by 0. Nevertheless, to be able to cover all the waves consistently we advise you to drop both missing and 0 values of *dpi*.

**Warning!** When you start working with smaller sub-samples, dropping observations may significantly affect your results if dropped observations all belong to one group that is central to your analysis (e.g., older immigrants, or low-educated blue-collar workers). Be careful about what you are doing. Understand your data.

- To equalise income, divide the total household income by the value of the equivalence scale for each observation. To generate LIS equivalised income:

$$ey = dpi / \text{SQRT}(d4) ;$$

- Be careful when using weights. Make sure that the weight matches your unit of analysis. Weigh by *hweight* for variables which are intrinsically at the household level (e.g., *dpi*) and by *hweight\*d4* (to account for household size) for variables that are conceptually meaningful at the person level (e.g., per capita and equivalised income).

## Program

```
OPTIONS NONOTES NOSOURCE NOFMterr NODATE NONUMBER NOCENTER LABEL LS=MAX  
PS=MAX;
```

```
/** MACRO DECLARATION **/
```

```
%MACRO avg ;
```

```
    TITLE &titre ;
```

```
    PROC MEANS DATA=eq MEAN;
```

```
        &where ;
```

```
        &by ;
```

```
        WEIGHT &poids ;
```

```
        VAR    &varlist ;
```

```
    RUN;
```

```
%MEND avg ;
```

```
/** START PROGRAM **/
```

```
DATA eq ;
```

```
    SET &fi00h (KEEP=casenum hweight d4 dpi);
```

```
    * Cleaning data;
```

```
    IF dpi in (. 0) THEN DELETE;
```

```
    * Income per capita;
```

```
    ypc = dpi / d4;
```

```
    * Equivalised income;
```

```
    ey = dpi /SQRT(d4);
```

```
    * Individual weight;
```

```
    wt=hweight * d4 ;
```

```
RUN;
```

```
PROC SORT DATA=eq;
```

```
    BY d4;
```

```
RUN;
```

```
%LET where ;
```

```
%LET by ;
```

```
%LET poids = hweight ;
```

```
%LET varlist = dpi ;
```

```
%LET titre = "TOTAL POPULATION - var: &varlist" ;
```

```
%avg
```

```
%LET poids = wt ;
```

```
%LET varlist = ypc ey ;
```

```
%avg
```

```

%LET where = WHERE d4 LE 7 ;
%LET by = BY d4 ;
%LET poids = hweight;
%LET varlist = dpi ;
%LET titre = "BY HOUSEHOLD SIZE - var: &varlist" ;
%avg
%LET poids = wt ;
%LET varlist = ypc ey ;
%avg

```

## **Results**

	<b>Total income</b>	<b>Per capita income</b>	<b>Equivalised income</b>
Average income for all households	<i>144 891</i>	<i>67 338</i>	<i>105 377</i>
Average income for households:			
- with 1 member	<i>77 475</i>	<i>77 475</i>	<i>77 475</i>
- with 2 members	<i>160 425</i>	<i>80 212</i>	<i>113 438</i>
- with 3 members	<i>198 849</i>	<i>66 283</i>	<i>114 806</i>
- with 7 members	<i>234 376</i>	<i>33 482</i>	<i>88 586</i>
What is the relationship between income and household size?	<i>Positive</i>	<i>Negative</i>	<i>No clear pattern</i>

## **Comments**

- Total household income obviously increases with household size, whereas per capita household income generally decreases. Neither of these two measures is appropriate to compare the well-being of households of different sizes.
- We use an equivalence scale because we believe that there are economies of scale in a household. Therefore, the marginal income needed decreases as the household size grows. As a result, equivalised income becomes independent from the household size, and we can compare different households.
- Note that when calculating statistics, it is always important to check for the cell size: the average income measures by number of household members, may be based on very few observations when the household size increases. (In this specific case, the number of households with more than 8 members drops to less than 30 observations, so that no sound conclusion can be taken for that group of households.)